

# 2

## SPREADSHEET FUNCTIONS AND MACROS

### Objectives

- Learn how to use the Paste Function menu on your spreadsheet to carry out a set of mathematical operations.
- Become familiar with three types of spreadsheet functions: standard functions, nested functions, and array functions.
- Practice using a variety of common spreadsheet functions.
- Develop and run a macro.

### INTRODUCTION

Mathematical functions describe natural phenomena in the form of an equation, relating one variable to another. In Exercise 1, you learned about linear, exponential, and power mathematical functions.

In this exercise, the “function” under discussion is quite different. **Spreadsheet functions** are formulae that have been written by a computer programmer to perform mathematical and other operations (see pp. 9–12). Your spreadsheet package likely has over 100 functions available for your use. These functions can make modeling easier for you, and you will use them extensively throughout this book.

#### **Standard Functions**

As an introduction to spreadsheet functions, let’s suppose that there are eight people in an elevator. The names of the eight individuals and their weights are given in Figure 1.

	A	B
1	<b>Individual</b>	<b>Weight (lbs)</b>
2	Tim	180
3	Anne	135
4	Pat	200
5	Donna	140
6	Kathleen	142
7	Joe	190
8	Mike	176
9	Tansy	135
10	<b>SUM =&gt;</b>	

Figure 1

Imagine that the elevator can hold a maximum of 1,500 pounds, and that a ninth person would like to get on. Would the addition of a ninth person exceed the 1,500-pound safety limit? To answer this question, we need to know how much the eight people in the elevator collectively weigh, and the weight of the ninth person. We could add cells B2–B9 to determine how much the eight people weigh. If we entered a mathematical formula in cell B10 to compute this, the formula reads  $=B2+B3+B4+B5+B6+B7+B8+B9$ . The result is 1,298 pounds. The more complicated a formula becomes, however, the more likely it is that you will make a mistake in entering it. This is where spreadsheet functions come into play. Instead of entering  $=B2+B3+B4+B5+B6+B7+B8+B9$  in cell B10, we can use the **SUM** spreadsheet function and have the spreadsheet do the work.

To enter a spreadsheet function, first select the cell in which you want the function to be computed (in this case, cell B10). Then you can do either of one of two things. You can use the Paste Function button  $f_x$  on your toolbar (indicated in Figure 2), or you can open Insert | Function to guide you through entering a function. Either way, the dialog box will appear as shown in Figure 2.




Figure 2

Look at the column on the left side of the dialog box. It asks what kinds of function category you want to examine. You could choose to look at the most recently used functions, or you can look at all the available functions, or you can check out the functions in a specific category, such as financial functions, statistical functions, and so on. If you choose All as a Function category, you'll see every function available in your spreadsheet package, listed in alphabetical order.

In Figure 2, we selected the Most Recently Used function category, so a list of the most recently used functions appears in the right side of the dialog box. Note that the function **SUM** is selected, and the program displays a brief description of the function at the bottom of the box: "Adds all the numbers in a range of cells." Click OK when you've got the function you want (in this case, the **SUM** function). Another box will then appear, called the **formula palette** (Figure 3). Each function has its own formula palette. You are asked to enter the addresses of the cells you wish to sum in the **SUM** formula palette. You can enter cell B2 as Number 1, cell B3 as Number 2, cell B4 as Number 3, and so on. Or you can type in the range **B2:B9** as Number 1 and the spreadsheet will recognize that the entire range of cells is to be added. When you are finished, click the OK box, or click on the green check-mark button to the left of the formula bar. If you




Figure 3

change your mind and decide to abandon the formula entry, click on the red × button to the left of the formula bar.

There are two handy features in a formula palette that you should note.

- First, notice the small figure with a red arrow pointing upward and leftward (located to the right of the blank space labeled Number 1). If you click on this arrow, the dialog box will shrink, exposing your spreadsheet so that you can use your mouse to select the range of cells you want to add. This is handy because you don't have to type in the cell references—just point and click on the appropriate cells. After you've selected the cells you want to add (in this case, use your mouse to highlight cells B2–B9), click on the arrow again and the SUM dialog box will reappear.
- The second handy feature of all Paste Function dialog boxes is “Help” information, accessed by clicking on the question mark located at the bottom-left corner of the window. If you don't know how the function works, clicking on the question mark will provide additional information.

Once you have entered all the necessary data and pressed OK, the spreadsheet will return the answer in cell B10. Although the spreadsheet displays the answer (1,298) in cell B10, the formula bar shows that the cell really contains the function =SUM(B2:B9). Note that the spreadsheet automatically inserted an equal sign before the function name, alerting the spreadsheet that a function is being used (Figure 4).




Figure 4

### Nested Functions

In some cases, you may need to perform more than one function, “nesting” one function inside another to give you the result you want. Returning to our elevator example, suppose that a ninth person, Peter, would like to board the elevator. He weighs 200 pounds. We want to enter a formula in cell B13 to determine whether he can safely board or not. If the total weight is less than 1,500 pounds, he can safely board. If the total weight is more than 1,500 pounds, he cannot safely board. We can use an **IF** function in cell B13 to carry out the operation and return the word “yes” if he can board or “no” if he cannot board (Figure 5).

	A	B
1	<b>Individual</b>	<b>Weight (lbs)</b>
2	Tim	180
3	Anne	135
4	Pat	200
5	Donna	140
6	Kathleen	142
7	Joe	190
8	Mike	176
9	Tansy	135
10	<b>SUM =&gt;</b>	1298
11		
12	Peter	200
13	<b>SAFELY BOARD?</b>	

Figure 5

As with the **SUM** function, you can use the Paste Function menu and then search for and select the **IF** function (Figure 6). You will notice at the bottom of the dialog box the words **IF(logical\_test,value\_if\_true,value\_if\_false)**. This is the syntax for the **IF** formula, and it provides the “rules” for entering an **IF** function. You should also see a brief description of the function that tells you the function “returns one value if a condition you specify evaluates to TRUE and another value if it evaluates to FALSE.”




Figure 6

For our example, we want to determine whether the total weight is less than or equal to 1,500 pounds. This is the logical test. If the logical test is TRUE, we want the word YES to be returned (he can safely board). If the logical test is FALSE, we want the word NO to be returned (he should not board). The formula palette for the IF function is shown in Figure 7.




Figure 7

The logical test requires that we sum the weights of the original eight individuals in cells B2–B9 and the weight of the ninth individual (cell B12) and determine whether the sum is less than 1,500. Because the logical test (IF function) contains the SUM function, it is called a **nested function**. To nest the SUM function within the IF function, place your cursor within the Logical\_test box. Then select the down arrow to the left of the formula bar. A list of functions appears. Search for the SUM function and click on it, and the SUM function palette will appear as shown in Figure 3. Enter the cell range B2:B9 as Number 1, and cell B12 as number 2. Instead of clicking OK when you are finished with the SUM function, click on the word IF on the formula bar; you will be returned to the IF formula palette and can complete the IF function entries.

Notice that the formula palette in Figure 7 displays the result of the logical test (TRUE) and the formula result (YES), indicating that Peter can board the elevator safely. The final function in cell B13 reads =IF(SUM(B2:B9,B12)<1500,"YES","NO"). When functions are nested within other functions, the spreadsheet will compute the answer to the “nested” functions (in this case, SUM) first and then will complete the outer functions.

### Array Formulae

Functions such as SUM perform a calculation and generate a result in a single cell. An **array formula**, on the other hand, can perform multiple calculations, returning either a single result or multiple results. Array formulae act on two or more sets of values known as “array arguments.”

You create array formulae in the same way you create other formulae, with a few major exceptions. First, instead of selecting a single cell to enter a formula, you need to select a *series* of cells, then enter an array formula. And second, instead of pressing OK after you have completed the entries in the function palette, you press <Control>+<Shift>+<Enter> (on Windows-based machines) or <Apple>+<Return> (on Macs) to enter the formula for all of the cells you have selected.

Let’s consider a new example. Suppose you want to construct a frequency distribution from the data in Figure 8. The weights (in grams) for 10 individuals are given in column B. Suppose you want to count the number of individuals that are 1 gram, 2 grams, 3 grams, 4 grams, and 5 grams. You could use the FREQUENCY function, which is an array formula to generate frequency data quickly.




Figure 8

The column labeled “Bins” in Figure 8 tells Excel how you want your data grouped. You can think of a bin as a bucket in which specific numbers go. The bins may be very small (hold a single or a few numbers) or very large (hold a large set of numbers). In this case, the numbers 1 through 5 represent the bins, and each bin “holds” just a single number. The task now is to have the spreadsheet count the number of individuals in each bin and return the answer in cells D2–D6. Because the frequency function is an array function, we need to select cells D2–D6 (rather than a single cell) *before* using the  $f_x$  button to summon the **FREQUENCY** formula.

The **FREQUENCY** formula palette will appear (Figure 8) and will guide you through the entries. The Data\_array is simply the data you want to summarize, given in cells B2:B11. The Bins\_array is cells C2:C6. Instead of clicking OK, press <Control>+<Shift>+<Enter> on Windows machines, or <Apple>+<Return> on Macs, and the spreadsheet will return your frequencies. If we examine the formulae in cells D2–D6, every cell will have the formulae {=FREQUENCY(B2:B11,C2:C6)}. The { } symbols indicate that the formula is part of an array.

Typically, frequency data are depicted graphically as shown in Figure 9. If you change the data set in some way, the spreadsheet will automatically update the frequencies. If for some reason you get “stuck” in an array formula, just hit the Escape key and start again.




Figure 9

### MACROS

As noted in the Introduction (p. 16), a **macro** is a miniature program that you build for yourself in order to run a sequence of spreadsheet actions. Typing and mouse-clicking

your way through a long series of commands over and over is time-consuming, boring, and error-prone. A macro allows you to achieve the same results with a single command.

You record a macro using Excel's built-in macro recorder. Start the recorder by opening Tools | Macro | Record New Macro (Figure 10).




Figure 10

The program will prompt you to name the macro and create a keyboard shortcut. Then a small window will appear with the macro recorder controls (Figure 11). If this button does not appear, go to View | Toolbars | Stop Recording, and the Stop Recording figure will appear. The square on the left side of the button is the Stop Recording button. When you press this square, you will stop recording your macro. The button on the right is the Relative Reference button. By default this button is not selected so that your macro recorder assumes that the cell references you make in the course of developing your macro are absolute. In other words, if you select cell A1 as part of a macro, Excel will interpret your keystroke as cell \$A\$1. There are cases (for example, the Survival Analysis exercise) in which you will want to select the relative reference button as you record your macro.

Once you have entered the macro name and shortcut key, the spreadsheet will record every action you take. Carry out the entire sequence of operations you want the macro




Figure 11

to do, and then press the Stop Recording button in the macro recorder control window. From this point on, Excel will mimic that entire sequence of actions whenever you press the keyboard shortcut or issue the macro command.

## PROCEDURES

Now that you have been introduced to simple functions, nested functions, arrays, and macros, it's time to put them into practice. The following instructions will introduce you to some 20 commonly used spreadsheet functions. As in Exercise 1, the left-hand column of instructions gives rather generic directions, and the right-hand column gives a step-by-step breakdown of these and explanatory comments or annotations. Try to think through and carry out the instructions in the left-hand column before referring to the right-hand column for confirmation. It's tempting to jump to the right hand column for the answers and explanation, but you will learn a lot more about using spreadsheet functions if you attempt it on your own. As always, **save your work frequently to disk**.

### INSTRUCTIONS

A. Set up the spreadsheet.

1. Open a new spreadsheet and enter headings as shown in Figure 12.

### ANNOTATION

	A	B	C
1	<b>Spreadsheet Functions and Macros</b>		
2			
3	<b>Individual</b>	<b>Height (cm)</b>	
4	1	12	
5	2	2	
6	3	8	
7	4	20	
8	5	3	
9	6	5	
10	7	12	
11	8	6	
12	9	4	
13	10	9	
14	11	7	
15	12	4	
16	13	1	
17	14	7	
18	15	7	
19	16	10	
20	17	1	
21	18	3	
22	19	2	
23	20	4	

Figure 12

2. Set up a linear series from 1 to 20 in cells A4–A23.

3. Enter the heights for the 20 individuals in cells B4–B23 as shown.

**B. Compute simple functions.**

1. Set up new headings as shown in Figure 13.

We will consider a sample of 20 individuals and their heights. Enter 1 in cell A4. Enter =1+A4 in cell A5. Select cell A5 and copy it down to cell A23.

These are the actual data, so just type in the numbers as shown in Figure 12.

In this section, you'll use 11 standard spreadsheet functions to compute various things, like the average height of the 20 individuals. For all functions, use the Paste Function menu (the Paste Function button,  $f_x$ , or open Insert | Function) to locate the appropriate function, review the function's formula palette, and complete the entries. You can double-check your results with ours at the end of the section.

	D	E
3	<b>Simple functions</b>	
4		
5	<b>Count</b>	
6	<b>Sum</b>	
7	<b>Average</b>	
8	<b>Median</b>	
9	<b>Mode</b>	
10	<b>Min</b>	
11	<b>Max</b>	
12	<b>Stdev</b>	
13	<b>4th large</b>	
14	<b>Rand</b>	
15	<b>Randbetween</b>	

Figure 13

2. In cell E5, use the **COUNT** spreadsheet function to count the total number of individuals in the sample.

3. In cells E6–E12, use the spreadsheet functions **SUM**, **AVERAGE**, **MEDIAN**, **MODE**, **MIN**, **MAX**, and **STDEV** to compute basic descriptive statistics for the population.

The **COUNT** function counts the number of cells that contain numbers. In this case, you want to count the number of times that a number is contained in cells B4–B23. Select the **COUNT** function from the Paste Function menu and compute this result. After you are finished, cell E5 should display the number 20, and its formula should be =COUNT(B4:B23).

For each formula, use the Paste Function menu and read through the information on the formula palette carefully. If you are unsure of the kind of information a statistic provides, click on the question mark on the bottom-left corner of the formula palette. After you have finished, the formulae in your spreadsheet should look like Figure 14, except that instead of seeing the formula in cells E5–E12, the *answers* to each formula will be displayed.

	D	E
3	<b>Simple functions</b>	
4		
5	<b>Count</b>	=COUNT(B4:B23)
6	<b>Sum</b>	=SUM(B4:B23)
7	<b>Average</b>	=AVERAGE(B4:B23)
8	<b>Median</b>	=MEDIAN(B4:B23)
9	<b>Mode</b>	=MODE(B4:B23)
10	<b>Min</b>	=MIN(B4:B23)
11	<b>Max</b>	=MAX(B4:B23)
12	<b>Stdev</b>	=STDEV(B4:B23)

Figure 14

4. In cell E13, use the **LARGE** function to compute the fourth largest height.

5. In cell E14, use the **RAND** formula to generate a random number between 0 and 1.

6. In cell E15, use the **RANDBETWEEN** function to generate a random number between 1 and 20.

7. Press F9, the Calculate key, to generate new random numbers in cells E14 and E15.

8. Save your work.

### *C. Compute multistep and nested functions.*

The **LARGE** function returns the  $k$ th largest value in a range of cells. In this case, the range of cells is B4–B23 (Figure 12), and  $k = 4$ . Your formula should read **=LARGE(B4:B23,4)**, and the answer should be 10.

You will use the **RAND** function in many of the exercises in this book. This function has the form **=RAND()**. The ( and ) are open and closed parentheses; you do not need to put anything inside them.

The **RANDBETWEEN** function generates a random integer between two specified values. The bottom value is the lowermost integer that can be randomly selected (1), and the top value is the uppermost integer that can be randomly selected (20). This function could be used to randomly select an individual from the population. The formula in cell E15 should read **=RANDBETWEEN(A4,A23)** or **=RANDBETWEEN(1,20)**.

*Note:* If your spreadsheet doesn't have the **RANDBETWEEN** function, you can enter the nested functions **=ROUNDUP(RAND()\*20,0)**. This will generate a random number between 0 and 1, multiply it by 20, and round it up to the nearest zero decimal places (i.e., to the nearest integer).

The Calculate key in Windows is the F9 key, located at the top of your keyboard.\* When this button is pushed, the spreadsheet will recalculate all of the formulae in the spreadsheet. For random numbers, such as those generated by the **RAND** or **RANDBETWEEN** functions, a new random number will be generated when the spreadsheet is calculated.

Verify this by examining the results in cells E14–E15 each time you press F9.

Now we will turn to nested functions and multi-step functions. Multi-step functions are actually standard functions like **SUM**, **MIN**, and **MAX**, but there are more entries involved in the formula palette. A function is *nested* if it uses more than one function to complete the calculations.

\*The F9 function key will work on Macintosh machines provided the Hot Function Key option in the Keyboard Control dialog box is turned OFF. If the F9 key does not work on your Mac, use the alternative, **⌘+=**.

1. Set up new headings as shown in Figure 15.

	F	G
3	<b>Multi-step and nested functions</b>	
4		
5	<b>Countif</b>	
6	<b>And</b>	
7	<b>Or</b>	
8	<b>Concatenate</b>	
9	<b>Vlookup</b>	
10	<b>Norminv</b>	
11	<b>Round</b>	
12	<b>If</b>	
13	<b>Random height</b>	

Figure 15

2. In cell G5, use the **COUNTIF** formula to count the number of times the modal value (given in cell E9) occurs.

3. In cell G6, use the **AND** function to determine if the value in cell B4 = 12 and the value in cell B5 = 2.

4. In cell G7, use the **OR** function to determine if the value in cell B5 is either 1 or 2.

5. In cell G8, use the **CONCATENATE** function to join the text in cell F6 with the text in cell F7.

6. In cell G9, use the **VLOOKUP** function to return the height of individual 1.

7. In cell G10, use the **NORMINV** function to draw a random data point from a distribution whose mean is given in cell E7, and whose standard deviation is given in cell E12.

We use the **COUNTIF** formula extensively. It counts the number of times a specific value occurs within a range of cells. Your formula should read `=COUNTIF(B4:B23,E9)` in cell G5, and your result should be 3, indicating that 3 individuals are 4 cm. in height.

The **AND** function returns the word **TRUE** or **FALSE**. It returns the word **TRUE** if all of the arguments in the formula are true (cell B4 = 12 *and* cell B5 = 2). If either condition is not true, the spreadsheet returns the word **FALSE**. Your result should be **TRUE**.

The **OR** function is similar to the **AND** function in that it returns the word **TRUE** or **FALSE**. It returns the word **TRUE** if any of the arguments in the formula are true (cell B5 = 1 *or* cell B5 = 2). Your result should be **TRUE**.

The **CONCATENATE** function joins several text strings into a single text string. The formula `=CONCATENATE(F6,F7)` should return the word "AndOr." This doesn't mean anything, but serves to illustrate the function. We will use this function in many of the genetics exercises. (The formula `=F6&F7` would generate the same result.)

The **VLOOKUP** function searches in the first column of a table for a value that you specify and returns the value of the corresponding cell in a different column. The **VLOOKUP** function needs three pieces of information: the value you want to find in the first column of the table, the cells that define the table (the upper-left and lower-right cells of the table), and the number of the column in the table that holds the information you want the formula to return. The formula `=VLOOKUP(1,A4:B23,2)` looks for the number 1 in the first column of the table defined by cells A4–B23, and it returns the value of the cell from the same row in the second column. In our spreadsheet, this formula returns the height of individual 1.

The **NORMINV** function is used extensively throughout the book, and is described more fully in Exercise 3, "Statistical Distributions." Since here you will use the **RAND** function within the **NORMINV** function, this is a *nested* formula. Generally speaking, for a set of normally distributed data, the function will generate a data value if you specify a probability associated with a normal curve. The function in cell G10 should read `=NORMINV(RAND(),E7,E12)`. In this case, we will first generate a random probability between 0 and 1. This probability will be applied to a normal distribution whose

8. In cell G11, use the **ROUND** function to round cell G10 to 0 decimal places.

9. In cell G12, use an **IF** function to return the number 0 if cell G11 is a negative number.

10. In cell G13, use the **VLOOKUP** function to look up the height of the randomly selected individual listed in cell E15.

11. Save your work.

*D. Utilize an array function.*

1. Set up new headings as shown in Figure 16.

2. Select cells I6–I9; then use the **FREQUENCY** function to generate frequency data of heights in the population. Use the bins in cells H6–H9.

mean is given in cell E7 and whose standard deviation is given in cell E12. The spreadsheet will then return the data value associated with that probability. Note when you press F9, the Calculate key, a new random number is computed, and thus a new random data point from the normal distribution is drawn. Also note that occasionally a negative number will appear. This is because the mean is close to 0 (6.35) and the standard deviation is quite large (4.65), so some of the data points within this distribution are below 0.

Your formula should read **=ROUND(G10,0)**. Once you are familiar with this function, you may find yourself typing it in by hand.

Your formula should read, **=IF(G11<0,0,G11)**. This tells the spreadsheet to evaluate the value in cell G11; if the number is < 0, return a 0; otherwise, return the number given in cell G11. This formula will prevent the spreadsheet from generating negative heights.

Your formula should read **=VLOOKUP(E15,A4:B23,2)**.

	H	I
3	<b>Array function</b>	
4		
5	<b>"Bin"</b>	<b>Frequency</b>
6	5	
7	10	
8	15	
9	>15	
10		

**Figure 16**

Remember that the **FREQUENCY** function is an array function. For this example, each bin "holds" several numbers. The bin labeled 5 holds heights that are up to and including 5 cm. The bin labeled 10 holds heights that are 6, 7, 8, 9, and 10 cm. Don't forget that to enter an array function such as the **FREQUENCY** function, you must press <Control>+<Shift>+<Enter> to generate a proper result. Cells I6–I9 should have the formula **{=FREQUENCY(B4:B23,H6:H9)}**.

3. Create a frequency histogram of the data in cells I6–I9. Label your axes fully (Figure 17).




Figure 17

Your spreadsheet should now look as shown in Figure 18. Note that you will likely have different values in cells E14–E15, G10–G11, and G13 because random numbers are used to generate the results shown.

4. Double-check your results.

	D	E	F	G	H	I
3	Simple functions		Multi-step and nested functions		Array function	
4						
5	Count	20	Countif	3	"Bin" Frequency	
6	Sum	127	And	TRUE	5	10
7	Average	6.35	Or	TRUE	10	7
8	Median	5.5	Concatenate	AndOr	15	2
9	Mode	4	Vlookup	12	>15	1
10	Min	1	Norminv	8.418740538		20
11	Max	20	Round	8		
12	Stdev	4.648429276	If	8		
13	4th large	10	Random height	4		
14	Rand	0.498679379				
15	Randbetween	20				

Figure 18

5. Save your work.

*E. Write a macro to randomly select individuals from the population.*

Now we will write a macro to randomly select an individual from the population, and we will record its height in column K. We will do this for 20 samples. Remember that you generated a random number between 1 and 20 in cell E15. You also looked up this randomly selected individual’s height with the **VLOOKUP** function in cell G13. In our macro, we will press F9 to generate a new randomly selected individual, then we will copy the value in cell G13 into cell K5. We will repeat the process for the second sample, but we will record the height of the randomly selected individual in cell K6 (and so on).

1. Set up new headings as shown in Figure 19.

	J	K
3	<b>Macro</b>	
4	<b>Sample</b>	<b>Height</b>
5	1	
6	2	
7	3	
8	4	
9	5	
10	6	
11	7	
12	8	
13	9	
14	10	
15	11	
16	12	
17	13	
18	14	
19	15	
20	16	
21	17	
22	18	
23	19	
24	20	
25	<b>average</b>	

Figure 19

2. Write a macro to record the heights of 20 randomly sampled individuals from the population.

There are many ways you can construct a macro to complete the task; here is one suggestion.

- Open Tools | Options | Calculation, and set the Calculation key to manual.
- Open Tools | Macro | Record New Macro. A dialog box will appear.
- Enter in a macro name (such as Sample) and a shortcut key (such as <Control>+<t>).
- If the Stop Recording button does not appear, open View | Toolbars | Stop Recording. You should now see the Stop Recording toolbar on your spreadsheet. The filled square on the left is the Stop Recording button. Press this button with your mouse when you are finished recording your macro.
- Press F9, the calculate key, to generate a new randomly selected individual.
- Select cell G13, the height of the randomly selected individual, and open Edit | Copy.
- Select cell K4, the top row of the height column.
- Open Edit | Find. A dialog box will appear (Figure 21).




Figure 20

- Leave the box labeled Find What blank, and select the Search By Columns option. Click on Find Next, then on Close. Your cursor should have moved down to the next empty cell on your spreadsheet.
- Open Edit | Paste Special, and select the Paste Values option. Click OK.
- Click on the Stop Recording button.

That's all there is to it. Now when you press the shortcut key, <Control>+<t>, the spreadsheet will repeat the steps in the macro automatically. Run your macro until you have obtained the heights of 20 randomly sampled individuals. (Note that with this process, the same individuals can be sampled more than once.)

You can view the code that the spreadsheet "wrote" as a result of your keystrokes by going to Tools | Macros | Macro. Select the macro name of interest, and click on Edit. The Visual Basic code will be revealed. When you are finished, click on the x button in the upper-right hand corner of the spreadsheet to close the Visual Basic code. You will be returned to your original spreadsheet.

3. Save your work.

You may want to switch your calculation key back to automatic; otherwise, you must press F9 any time you want your spreadsheet to calculate values.

## QUESTIONS

1. Explore the formulae used in the exercise by changing some of the heights of the individuals. For example, change cell B5 from 2 to 1. How does this change affect the outcome of the **AND** and **OR** functions? Change other values in the data set as well. How do your changes affect the frequency distribution of the data?
2. Click on the Paste Function button,  $f_x$ , and select the function category ALL. A list of all functions is displayed on the right-hand side of the Paste Function dialog box. Click on a function that looks interesting, and notice the description of the function that appears in the lower portion of the dialog box. Select three functions that were *not* used in this exercise and explore how each function works. Choose functions that are likely to be relevant to the data set in the exercise.